Task

Write a boolean function that tells whether a given integer is prime.



Remember that   1   and all non-positive numbers are not prime.

Use trial division.

Even numbers greater than   2   may be eliminated right away.

A loop from   3   to   √ n    will suffice,   but other loops are allowed.



Related tasks

   count in factors
   prime decomposition
   AKS test for primes
   factors of an integer
   Sieve of Eratosthenes
   factors of a Mersenne number
   trial factoring of a Mersenne number
   partition an integer X into N primes
   sequence of primes by Trial Division


get "libhdr"
 
let sqrt(s) = 
    s <= 1 -> 1,
    valof
$(  let x0 = s >> 1
    let x1 = (x0 + s/x0) >> 1
    while x1 < x0 
    $(  x0 := x1
        x1 := (x0 + s/x0) >> 1
    $)
    resultis x0
$)
 
let isprime(n) =
    n < 2       -> false,
    (n & 1) = 0 -> n = 2,
    valof
$(  for i = 3 to sqrt(n) by 2
        if n rem i = 0 resultis false
    resultis true
$)
 
let start() be
$(  for i=1 to 100
        if isprime(i) then writef("%N ",i)
    wrch('*N')
$)